1.6 Singulární kvadriky

```
> restart;
> with(LinearAlgebra):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Determinant Δ matice kvadriky - velký determinant

```
> Delta:=Determinant(K);
```

$$\begin{aligned} \Delta := {} & a(1,1)\,a(2,2)\,a(3,3)\,a(4,4) - a(1,1)\,a(2,2)\,a(3,4)^2 \\ & + 2\,a(1,1)\,a(2,3)\,a(3,4)\,a(2,4) - a(1,1)\,a(2,3)^2\,a(4,4) - a(1,1)\,a(3,3)\,a(2,4)^2 \\ & - a(1,2)^2\,a(3,3)\,a(4,4) + a(1,2)^2\,a(3,4)^2 - 2\,a(1,2)\,a(2,3)\,a(3,4)\,a(1,4) \\ & + 2\,a(1,2)\,a(2,3)\,a(1,3)\,a(4,4) - 2\,a(1,2)\,a(2,4)\,a(1,3)\,a(3,4) \\ & + 2\,a(1,2)\,a(2,4)\,a(3,3)\,a(1,4) + 2\,a(1,3)\,a(2,2)\,a(3,4)\,a(1,4) \\ & - a(2,2)\,a(1,3)^2\,a(4,4) + a(1,3)^2\,a(2,4)^2 - 2\,a(1,3)\,a(2,4)\,a(2,3)\,a(1,4) \\ & - a(2,2)\,a(3,3)\,a(1,4)^2 + a(2,3)^2\,a(1,4)^2 \end{aligned}$$

Rovnice kvadriky:

```
> Kv:=expand(X.K.Transpose(X))=0;
```

$$\begin{aligned} Kv := {} & x^2\,a(1,1) + 2\,x\,y\,a(1,2) + 2\,x\,z\,a(1,3) + 2\,x\,a(1,4) + y^2\,a(2,2) + 2\,y\,z\,a(2,3) \\ & + 2\,y\,a(2,4) + z^2\,a(3,3) + 2\,z\,a(3,4) + a(4,4) = 0 \end{aligned}$$

Bod M (singulární bod kvadriky):

```
> M:=[m,n,p];
```

$$M := [m, n, p]$$

Parametrické rovnice přímky procházející bodem M:

```
> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Primka := [x = m + t\,u, y = n + t\,v, z = p + t\,w]$$

Rovnice průniku této přímky s kvadrikou:

```
> Kv1:=simplify(eval(Kv,Primka));
```

$$\begin{aligned} Kv1 := {} & 2\,a(1,1)\,m\,t\,u + 2\,a(1,2)\,m\,t\,v + 2\,a(1,2)\,t\,u\,n + 2\,a(1,2)\,t^2\,u\,v + 2\,a(1,3)\,m\,t\,w \\ & + 2\,a(1,3)\,t\,u\,p + 2\,a(1,3)\,t^2\,u\,w + 2\,a(2,2)\,n\,t\,v + 2\,a(2,3)\,n\,t\,w + 2\,a(2,3)\,t\,v\,p \\ & + 2\,a(2,3)\,t^2\,v\,w + 2\,a(3,3)\,p\,t\,w + a(4,4) + a(1,1)\,t^2\,u^2 + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p \end{aligned}$$

$$+2\,a(1,4)\,t\,u + a(2,2)\,t^2 v^2 + 2\,a(2,3)\,n\,p + 2\,a(2,4)\,t\,v + a(3,3)\,t^2 w^2 + 2\,a(3,4)\,t\,w + a(1,1)\,m^2 + 2\,a(1,4)\,m + a(2,2)\,n^2 + 2\,a(2,4)\,n + a(3,3)\,p^2 + 2\,a(3,4)\,p = 0$$

Koeficient B rovnice $At^2 + Bt + C = 0$ společných bodů kvadriky a přímky:

```
> B:=coeff(lhs(Kv1),t)/2;
```

$$B := a(1,1)\,m\,u + a(1,2)\,m\,v + a(1,2)\,u\,n + a(1,3)\,m\,w + a(1,3)\,u\,p + a(2,2)\,n\,v + a(2,3)\,n\,w + a(2,3)\,v\,p + a(3,3)\,p\,w + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w$$

```
> B:=collect(B,[u,v,w]);
```

$$B := (a(1,1)\,m + a(1,3)\,p + a(1,4) + a(1,2)\,n)\,u + (a(1,2)\,m + a(2,2)\,n + a(2,4) + a(2,3)\,p)\,v + (a(3,3)\,p + a(2,3)\,n + a(1,3)\,m + a(3,4))\,w$$

Soustava rovnic - podmínek pro určení bodu $M = [m, n, p]$ jako singulárního bodu kvadriky :

```
> r1:=sort(coeff(B,u),[m,n,p])=0; r2:=sort(coeff(B,v),[m,n,p])=0;
  r3:=sort(coeff(B,w),[m,n,p])=0;
  r4:=sort(simplify(coeff(lhs(Kv1),t,0)-m*lhs(r1)-n*lhs(r2)-p*lhs(
  r3)),[m,n,p])=0;
```

$$r1 := a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4) = 0$$
$$r2 := a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4) = 0$$
$$r3 := a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4) = 0$$
$$r4 := a(1,4)\,m + a(2,4)\,n + a(3,4)\,p + a(4,4) = 0$$

Matice MS a rozšířená matice MR této soustavy:

```
> MS:=linalg[genmatrix]({r1,r2,r3,r4},[m,n,p]);
  MR:=linalg[genmatrix]({r1,r2,r3,r4},[m,n,p],flag);
```

$$MS := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) \\ a(1,2) & a(2,2) & a(2,3) \\ a(1,3) & a(2,3) & a(3,3) \\ a(1,4) & a(2,4) & a(3,4) \end{bmatrix}$$

$$MR := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & -a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & -a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & -a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & -a(4,4) \end{bmatrix}$$

1) Hodnost matice kvadriky $h(K) = 3$

```
> M:=[0,0,0]:
> res1:=solve(eval({r1,r2,r3,r4},[m=0,n=0,p=0]),{a(1,4),a(2,4),a(3
  ,4),a(4,4)});
```

$$res1 := \{a(1,4) = 0,\ a(2,4) = 0,\ a(3,4) = 0,\ a(4,4) = 0\}$$

```
> assign(res1);
> Kv;
```

$$x^2\,a(1,1) + 2\,x\,y\,a(1,2) + 2\,x\,z\,a(1,3) + y^2\,a(2,2) + 2\,y\,z\,a(2,3) + z^2\,a(3,3) = 0$$

Uvažujme konkrétní kvadriku daných vlastností:

```
> a(1,1):=1: a(1,2):=-1: a(1,3):=2: a(2,2):=3: a(2,3):=-2:
  a(3,3):=1:
```

Odpovídající kvadrikou je kuželová plocha o rovnici:

```
> Kv1:=Kv;
```

$$Kv1 := x^2 - 2\,x\,y + 4\,x\,z + 3\,y^2 - 4\,y\,z + z^2 = 0$$

```
> Krivka:=eval(Kv1,z=-4);
```

$$Krivka := x^2 - 2\,x\,y - 16\,x + 3\,y^2 + 16\,y + 16 = 0$$

```
> K_y:=solve(Krivka,y);
```

$$K\_y := \frac{x}{3} - \frac{8}{3} + \frac{\sqrt{-2\,x^2 + 32\,x + 16}}{3},\ \frac{x}{3} - \frac{8}{3} - \frac{\sqrt{-2\,x^2 + 32\,x + 16}}{3}$$

```
> p1:=map(unapply,expand([0,0,0]+t*[x,K_y[1],-4]),x);
```

$$p1 := \left[x \to x\,t,\ x \to \frac{1}{3}\,x\,t - \frac{8\,t}{3} + \frac{1}{3}\sqrt{-2\,x^2 + 32\,x + 16}\;t,\ x \to -4\,t\right]$$

```
> p2:=map(unapply,expand([0,0,0]+t*[x,K_y[2],-4]),x);
```

$$p2 := \left[x \to x\,t,\ x \to \frac{1}{3}\,x\,t - \frac{8\,t}{3} - \frac{1}{3}\sqrt{-2\,x^2 + 32\,x + 16}\;t,\ x \to -4\,t\right]$$

```
> p1f:=j->plot3d(p1(j),t=-2..2,s=-1..-1,axes=frame);
```

$$p1f := j \to \mathrm{plot3d}(\mathrm{p1}(j),\ t = \text{-}2\,..\,2,\ s = \text{-}1\,..\,\text{-}1,\ axes = frame)$$

```
> p2f:=j->plot3d(p2(j),t=-2..2,s=-1..-1,axes=frame);
```

$$p2f := j \to \mathrm{plot3d}(\mathrm{p2}(j),\ t = \text{-}2\,..\,2,\ s = \text{-}1\,..\,\text{-}1,\ axes = frame)$$

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> primky:=plots[display](seq(p1f(k),k=-10..20,2),seq(p2f(k),k=-10.
  .20,2),orientation=[-60,48],scaling=constrained,view=[-18..18,-1
  0..10,-4..4]):
```

Grafické znázornění

Znázornění užitím příkazu **plots[implicitplot3d]**

```
> Plocha:=plots[implicitplot3d](Kv1,x=-18..18,y=-10..10,z=-4..4,gr
  id=[60,60,60],orientation=[-60,48],axes=frame,view=[-18..18,-10.
  .10,-4..4],scaling=unconstrained,style=patchcontour,color=COLOR(
  RGB,250/255,250/255,250/255),light=[90,-5,1,1,1],tickmarks=[3,3,
  3]):
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> plots[display](Plocha,primky);
```



Singulární kvadrikou s h(K) = 3 je i **válcová plocha** daná rovnicí:

```
> Pl:=a*x^2+2*b*x*y+c*y^2+2*d*x+2*e*y+f=0;
```

$$Pl := a\, x^2 + 2\, b\, x\, y + c\, y^2 + 2\, d\, x + 2\, e\, y + f = 0$$

```
> a:='a';
```

$$a := a$$

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} \mathrm{a}(1,1) & \mathrm{a}(1,2) & \mathrm{a}(1,3) & \mathrm{a}(1,4) \\ \mathrm{a}(1,2) & \mathrm{a}(2,2) & \mathrm{a}(2,3) & \mathrm{a}(2,4) \\ \mathrm{a}(1,3) & \mathrm{a}(2,3) & \mathrm{a}(3,3) & \mathrm{a}(3,4) \\ \mathrm{a}(1,4) & \mathrm{a}(2,4) & \mathrm{a}(3,4) & \mathrm{a}(4,4) \end{bmatrix}$$

```
> Kv:=expand(X.K.Transpose(X))=0;
```

$$Kv := x^2\, \mathrm{a}(1,1) + 2\, x\, y\, \mathrm{a}(1,2) + 2\, x\, z\, \mathrm{a}(1,3) + 2\, x\, \mathrm{a}(1,4) + y^2\, \mathrm{a}(2,2) + 2\, y\, z\, \mathrm{a}(2,3) + 2\, y\, \mathrm{a}(2,4) + z^2\, \mathrm{a}(3,3) + 2\, z\, \mathrm{a}(3,4) + \mathrm{a}(4,4) = 0$$

Porovnáme koeficienty rovnic Pl a Kv:

```
> KvPl:=collect(simplify(Pl-Kv),[x,y,z],distributed);
```

$$KvPl := f - \mathrm{a}(4,4) + (2\, e - 2\, \mathrm{a}(2,4))\, y + (2\, d - 2\, \mathrm{a}(1,4))\, x - 2\, z\, \mathrm{a}(3,4) + (-\mathrm{a}(1,1) + a)\, x^2 - z^2\, \mathrm{a}(3,3) + (-2\, \mathrm{a}(1,2) + 2\, b)\, x\, y - 2\, x\, z\, \mathrm{a}(1,3) - 2\, y\, z\, \mathrm{a}(2,3) + (-\mathrm{a}(2,2) + c)\, y^2 = 0$$

Dostaneme tak soustavu deseti rovnic pro členy a(i, j) matice K:

```
> r1:=coeff(lhs(KvPl),x,2); r2:=coeff(lhs(KvPl),y,2);
  r3:=coeff(lhs(KvPl),z,2); r4:=coeff(coeff(lhs(KvPl),x),y);
  r5:=coeff(coeff(lhs(KvPl),x),z);
  r6:=coeff(coeff(lhs(KvPl),y),z);
  r7:=coeff(coeff(coeff(lhs(KvPl),x),y,0),z,0);
  r8:=coeff(coeff(coeff(lhs(KvPl),y),x,0),z,0);
  r9:=coeff(coeff(coeff(lhs(KvPl),z),x,0),y,0);
  r10:=coeff(coeff(coeff(lhs(KvPl),x,0),y,0),z,0);
```

$$r1 := -a(1,1) + a$$

$$r2 := -a(2,2) + c$$

$$r3 := -a(3,3)$$

$$r4 := -2\,a(1,2) + 2\,b$$

$$r5 := -2\,a(1,3)$$

$$r6 := -2\,a(2,3)$$

$$r7 := 2\,d - 2\,a(1,4)$$

$$r8 := 2\,e - 2\,a(2,4)$$

$$r9 := -2\,a(3,4)$$

$$r10 := f - a(4,4)$$

```
> res:=solve({r1,r2,r3,r4,r5,r6,r7,r8,r9,r10},{a(1,1),a(1,2),a(1,3
  ),a(1,4),a(2,2),a(2,3),a(2,4),a(3,3),a(3,4),a(4,4)});
```

$res := \{a(1,1) = a, a(1,2) = b, a(1,3) = 0, a(1,4) = d, a(2,2) = c, a(2,3) = 0, a(2,4) = e,$
$a(3,3) = 0, a(3,4) = 0, a(4,4) = f\}$

Matice naší válcové plochy má tedy tvar:

```
> KVal:=map(x->eval(x,res),K);
```

$$KVal := \begin{bmatrix} a & b & 0 & d \\ b & c & 0 & e \\ 0 & 0 & 0 & 0 \\ d & e & 0 & f \end{bmatrix}$$

Hodnost je zřejmá:

```
> Rank(KVal);
```

$$3$$

Pro ilustraci použijeme konkrétní válcovou plochu (hyperbolická válcová plocha):

```
> a:=1: b:=-2: c:=1: d:=5: e:=-1: f:=3:
> Pl;
```

$$x^2 - 4\,x\,y + y^2 + 10\,x - 2\,y + 3 = 0$$

```
> ValPlocha:=plots[implicitplot3d](Pl,x=-8..8,y=-8..8,z=-8..8,orie
  ntation=[52,63],grid=[40,40,10],style=patchcontour,axes=frame,co
  lor=COLOR(RGB,250/255,250/255,250/255),light=[90,-5,1,1,1],tickm
  arks=[3,3,3],contours=5,scaling=constrained):
> plots[display](ValPlocha,orientation=[-66,53],view=[-8..8,-8..8,
```

```
-8..8],scaling=constrained);
```




```
>
```